NAG Library Routine Document

F07ATF (ZGEEQU)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07ATF (ZGEEQU) computes real diagonal scaling matrices D_R and D_C intended to equilibrate a complex m by n matrix A and reduce its condition number.

2 Specification

```
SUBROUTINE F07ATF (M, N, A, LDA, R, C, ROWCND, COLCND, AMAX, INFO)
INTEGER                 M, N, LDA, INFO
REAL (KIND=nag_wp)      R(M), C(N), ROWCND, COLCND, AMAX
COMPLEX (KIND=nag_wp) A(LDA,*)
```

The routine may be called by its LAPACK name ***zgeequ***.

3 Description

F07ATF (ZGEEQU) computes the diagonal scaling matrices. The diagonal scaling matrices are chosen to try to make the elements of largest absolute value in each row and column of the matrix B given by

$$B = D_R A D_C$$

have absolute value 1. The diagonal elements of D_R and D_C are restricted to lie in the safe range $(\delta, 1/\delta)$, where δ is the value returned by routine X02AMF. Use of these scaling factors is not guaranteed to reduce the condition number of A but works well in practice.

4 References

None.

5 Parameters

1: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: M ≥ 0.

2: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: N ≥ 0.

3: A(LDA, $*$) – COMPLEX (KIND=nag_wp) array *Input*

Note: the second dimension of the array A must be at least max(1, N).

On entry: the matrix A whose scaling factors are to be computed.

4: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F07ATF (ZGEEQU) is called.

Constraint: LDA $\geq \max(1, \mathrm{M})$.

5: R(M) – REAL (KIND=nag_wp) array *Output*

On exit: if INFO $= 0$ or INFO $>$ M, R contains the row scale factors, the diagonal elements of D_R. The elements of R will be positive.

6: C(N) – REAL (KIND=nag_wp) array *Output*

On exit: if INFO $= 0$, C contains the column scale factors, the diagonal elements of D_C. The elements of C will be positive.

7: ROWCND – REAL (KIND=nag_wp) *Output*

On exit: if INFO $= 0$ or INFO $>$ M, ROWCND contains the ratio of the smallest value of R(i) to the largest value of R(i). If ROWCND ≥ 0.1 and AMAX is neither too large nor too small, it is not worth scaling by D_R.

8: COLCND – REAL (KIND=nag_wp) *Output*

On exit: if INFO $= 0$, COLCND contains the ratio of the smallest value of C(i) to the largest value of C(i).

If COLCND ≥ 0.1, it is not worth scaling by D_C.

9: AMAX – REAL (KIND=nag_wp) *Output*

On exit: $\max \left|a_{ij}\right|$. If AMAX is very close to overflow or underflow, the matrix A should be scaled.

10: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO $= -i$, argument i had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0 and INFO $\leq$ M

Row $\langle$*value*$\rangle$ of A is exactly zero.

INFO $>$ M

Column $\langle$*value*$\rangle$ of A is exactly zero.

7 Accuracy

The computed scale factors will be close to the exact scale factors.

8 Parallelism and Performance

Not applicable.

9 Further Comments

The real analogue of this routine is F07AFF (DGEEQU).

10 Example

This example equilibrates the general matrix A given by

$$A = \begin{pmatrix} -1.34 + 2.55i & (0.28 + 3.17i) \times 10^{10} & -6.39 - 2.20i \\ -1.70 - 1.41i & (3.31 - 0.15i) \times 10^{10} & -0.15 + 1.34i \\ (2.41 + 0.39i) \times 10^{-10} & -0.56 + 1.47i & (-0.83 - 0.69i) \times 10^{-10} \end{pmatrix}.$$

Details of the scaling factors, and the scaled matrix are output.

10.1 Program Text

```
    Program f07atfe

!     F07ATF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: f06kcf, nag_wp, x02ajf, x02amf, x02bhf, x04dbf,   &
                             zdscal, zgeequ
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Real (Kind=nag_wp), Parameter    :: one = 1.0_nag_wp
      Real (Kind=nag_wp), Parameter    :: thresh = 0.1_nag_wp
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: amax, big, colcnd, rowcnd, small
      Integer                          :: i, ifail, info, j, lda, n
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: a(:,:)
      Real (Kind=nag_wp), Allocatable  :: c(:), r(:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: real
!     .. Executable Statements ..
      Write (nout,*) 'F07ATF Example Program Results'
      Write (nout,*)
      Flush (nout)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n
      lda = n
      Allocate (a(lda,n),c(n),r(n))

!     Read the N by N matrix A from data file

      Read (nin,*)(a(i,1:n),i=1,n)

!     Print the matrix A

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
      Call x04dbf('General',' ',n,n,a,lda,'Bracketed','1P,E10.2','Matrix A', &
        'Integer',rlabs,'Integer',clabs,80,0,ifail)

      Write (nout,*)

!     Compute row and column scaling factors

!     The NAG name equivalent of zgeequ is f07atf
      Call zgeequ(n,n,a,lda,r,c,rowcnd,colcnd,amax,info)
```

```
      If (info>0) Then
        If (info<=n) Then
          Write (nout,99999) 'Row ', info, ' of A is exactly zero'
        Else
          Write (nout,99999) 'Column ', info - n, ' of A is exactly zero'
        End If
      Else

!       Print ROWCND, COLCND, AMAX and the scale factors

        Write (nout,99998) 'ROWCND =', rowcnd, ', COLCND =', colcnd, &
          ', AMAX =', amax
        Write (nout,*)
        Write (nout,*) 'Row scale factors'
        Write (nout,99997) r(1:n)
        Write (nout,*)
        Write (nout,*) 'Column scale factors'
        Write (nout,99997) c(1:n)
        Write (nout,*)
        Flush (nout)

!       Compute values close to underflow and overflow

        small = x02amf()/(x02ajf()*real(x02bhf(),kind=nag_wp))
        big = one/small
        If ((rowcnd>=thresh) .And. (amax>=small) .And. (amax<=big)) Then
          If (colcnd<thresh) Then

!           Just column scale A
!           The NAG name equivalent of zdscal is f06jdf
            Do j = 1, n
              Call zdscal(n,c(j),a(1,j),1)
            End Do

          End If
        Else If (colcnd>=thresh) Then

!         Just row scale A
          Do j = 1, n
            Call f06kcf(n,r,1,a(1,j),1)
          End Do

        Else

!         Row and column scale A
          Do j = 1, n
            Call zdscal(n,c(j),a(1,j),1)
            Call f06kcf(n,r,1,a(1,j),1)
          End Do

        End If

!       Print the scaled matrix
        ifail = 0
        Call x04dbf('General',' ',n,n,a,lda,'Bracketed',' ','Scaled matrix', &
          'Integer',rlabs,'Integer',clabs,80,0,ifail)


      End If

99999 Format (1X,A,I4,A)
99998 Format (1X,3(A,1P,E8.1))
99997 Format ((1X,1P,7E11.2))
    End Program f07atfe
```

10.2 Program Data

```
F07ATF Example Program Data

  3                                                              :Value of N

 (-1.34D+00, 2.55D+00) ( 0.28D+10, 3.17D+10) (-6.39D+00,-2.20D+00)
 (-1.70D+00,-1.41D+00) ( 3.31D+10,-0.15D+10) (-0.15D+00, 1.34D+00)
 ( 2.41D-10, 0.39D-10) (-0.56D+00, 1.47D+00) (-0.83D-10,-0.69D-10)

                                                                :End of matrix A
```

10.3 Program Results

```
 F07ATF Example Program Results

 Matrix A
                            1                        2                        3
 1  ( -1.34E+00,  2.55E+00) (  2.80E+09,  3.17E+10) ( -6.39E+00, -2.20E+00)
 2  ( -1.70E+00, -1.41E+00) (  3.31E+10, -1.50E+09) ( -1.50E-01,  1.34E+00)
 3  (  2.41E-10,  3.90E-11) ( -5.60E-01,  1.47E+00) ( -8.30E-11, -6.90E-11)

 ROWCND = 5.9E-11, COLCND = 1.4E-10, AMAX = 3.5E+10

 Row scale factors
    2.90E-11   2.89E-11   4.93E-01

 Column scale factors
    7.25E+09   1.00E+00   4.02E+09

 Scaled matrix
                          1                    2                    3
 1  ( -0.2816,  0.5359) (  0.0812,  0.9188) ( -0.7439, -0.2561)
 2  ( -0.3562, -0.2954) (  0.9566, -0.0434) ( -0.0174,  0.1555)
 3  (  0.8607,  0.1393) ( -0.2759,  0.7241) ( -0.1642, -0.1365)
```